Chapter 9: Building and Using Decision Models

Statistics, Data Analysis, and Decision Modeling, Fifth Edition

James R. Evans

Models

- A model is an abstraction or representation of a real system, idea, or object.
- Models could be pictures, spreadsheets, or mathematical relationships
- Models contain data, uncontrollable variables, and decision variables



Decision Models – Outsourcing Example

- Decision models are models that can be used to understand, analyze, or facilitate making a decision

	A	B	
1	**Outsourcing Decision Model**		
2			
3	**Data**		
4			
5	**Manufactured in-house**		
6	Fixed cost	$	50,000
7	Unit variable cost	$	125
8			
9	**Purchased from supplier**		
10	Unit cost	$	175
11			
12	**Model**		
13			
14	Demand volume		1500
15			
16	Total manufacturing cost	$	237,500
17	Total purchased cost	$	262,500
18	Difference	$	(25,000)
19			
20	Decision	Manufacture	

	A	B
1	**Outsourcing Decision Model**	
2		
3	**Data**	
4		
5	**Manufactured in-house**	
6	Fixed cost	50000
7	Unit variable cost	125
8		
9	**Purchased from supplier**	
10	Unit cost	175
11		
12	**Model**	
13		
14	Demand volume	1500
15		
16	Total manufacturing cost	=B6+B7*B14
17	Total purchased cost	=B14*B10
18	Difference	=B16-B17
19		
20	Decision	=IF(B18<=0, "Manufacture", "Outsource")

Outsourcing Model

- Model components
  - F = fixed cost of in-house manufacturing
  - V = unit variable cost of in-house manufacturing
  - C = unit cost of outsourcing
  - D = demand volume
- Total Manufacturing Cost $= TMC = F + V * D$
- Total outsourcing cost = $TOC = C * D.$

Types of Decision Models

- Descriptive - describe relationships and provide information for evaluation
- Prescriptive (optimization models) - determine an optimal policy, that is, the best course of action that a decision maker should take to maximize or minimize some objective

Airline Pricing Model

	A	B
1	**Airline Pricing Model**	
2		
3	**Data**	
4	Airplane capacity	300
5	Fixed cost	$ 90,000
6	Demand function	
7	slope	-2.33
8	intercept	1900
9		
10	**Model**	
11		
12	Revenue	
13	Unit price	$ 500.00
14	Demand	733
15	Number of flights/day	3
16	Total Revenue	$366,666.67
17	Cost	
18	Fixed Cost	$270,000.00
19		
20	Profit	$96,666.67

	A	B
1	**Airline Pricing Model**	
2		
3	**Data**	
4	Airplane capacity	300
5	Fixed cost	90000
6	Demand function	
7	slope	=-7/3
8	intercept	1900
9		
10	**Model**	
11		
12	Revenue	
13	Unit price	500
14	Demand	=B8+B7*B13
15	Number of flights/day	=ROUNDUP(B14/B4,0)
16	Total Revenue	=B13*B14
17	Cost	
18	Fixed Cost	=B5*B15
19		
20	Profit	=B16-B18



Model Analysis

- What-If Analysis – evaluate how specific combinations of model inputs that reflect key model assumptions affect model outputs (often called sensitivity analysis).
- Excel tools
  - Data tables
  - Scenario manager
  - Goal seek



Data Tables

- Summarizes the impact of one or two inputs on a specified output
- Excel tools
  - One-way data tables
  - Two-way data tables

One Way Data Table

	A	B	C	D	E	F	G
1	**Outsourcing Decision Model**						
2					Fixed Costs	Difference	Decision
3	**Data**					$ (25,000)	Manufacture
4					$ 30,000	$ (45,000)	Manufacture
5	**Manufactured in-house**				$ 40,000	$ (35,000)	Manufacture
6	Fixed cost	$ 50,000			$ 50,000	$ (25,000)	Manufacture
7	Unit variable cost	$ 125			$ 60,000	$ (15,000)	Manufacture
8					$ 70,000	$ (5,000)	Manufacture
9	**Purchased from supplier**				$ 80,000	$ 5,000	Outsource
10	Unit cost	$ 175			$ 90,000	$ 15,000	Outsource
11					$ 100,000	$ 25,000	Outsource
12	**Model**						
13							
14	Demand volume	1500					
15							
16	Total manufacturing cost	$ 237,500					
17	Total purchased cost	$ 262,500					
18	Difference	$ (25,000)					
19							
20	Decision	Manufacture					

Column input cell

Two Way Data Table

	A	B	C	D	E	F	G	H	I	J	K
1	**Outsourcing Decision Model**										
2			Column input cell		Fixed Cost			Variable Cost			
3	**Data**				Manufacture	$ 100	$ 110	$ 120	$ 130	$ 140	$ 150
4					$ 30,000	Manufacture	Manufacture	Manufacture	Manufacture	Manufacture	Manufacture
5	**Manufactured in-house**				$ 40,000	Manufacture	Manufacture	Manufacture	Manufacture	Manufacture	Outsource
6	Fixed cost	$ 50,000			$ 50,000	Manufacture	Manufacture	Manufacture	Manufacture	Manufacture	Outsource
7	Unit variable cost	$ 125			$ 60,000	Manufacture	Manufacture	Manufacture	Manufacture	Outsource	Outsource
8			Row input cell		$ 70,000	Manufacture	Manufacture	Manufacture	Outsource	Outsource	Outsource
9	**Purchased from supplier**				$ 80,000	Manufacture	Manufacture	Manufacture	Outsource	Outsource	Outsource
10	Unit cost	$ 175			$ 90,000	Manufacture	Manufacture	Outsource	Outsource	Outsource	Outsource
11					$ 100,000	Manufacture	Outsource	Outsource	Outsource	Outsource	Outsource
12	**Model**										
13											
14	Demand volume	1500									
15											
16	Total manufacturing cost	$ 237,500									
17	Total purchased cost	$ 262,500									
18	Difference	$ (25,000)									
19											
20	Decision	Manufacture									

Scenario Manager

	Fixed Cost	Unit Variable Cost	Demand Volume
Best case	$40,000	$120	1,800
Worst case	$60,000	$140	1,000
Most likely case	$55,000	$125	1,500

Scenario Summary

	Current Values:	Best case	Worst case	Most likely case
Changing Cells:				
$B$6	$ 50,000	$ 40,000	$ 60,000	$ 55,000
$B$7	$ 125	$ 120	$ 140	$ 125
$B$14	1500	1800	1000	1500
Result Cells:				
$B$18	$ (25,000)	$ (59,000)	$ 25,000	$ (20,000)
$B$20	Manufacture	Manufacture	Outsource	Manufacture

Notes: Current Values column represents values of changing cells at time Scenario Summary Report was created. Changing cells for each scenario are highlighted in gray.

Goal Seek

- Find the value of an input that produces a known result within a spreadsheet
- Example: find the breakeven point in the outsourcing decision model

	A	B	
1	**Outsourcing Decision Model**		
2			
3	**Data**		
4			
5	**Manufactured in-house**		
6	Fixed cost	$	50,000
7	Unit variable cost	$	125
8			
9	**Purchased from supplier**		
10	Unit cost	$	175
11			
12	**Model**		
13			
14	Demand volume		1500
15			
16	Total manufacturing cost	$	237,500
17	Total purchased cost	$	262,500
18	Difference	$	(25,000)
19			
20	Decision	Manufacture	



Set cell is B18;
To value = 0;
By changing cell is B14

Optimization Models: Excel *Solver*



	A	B
1	**Airline Pricing Model**	
2		
3	**Data**	
4	Airplane capacity	300
5	Fixed cost	$ 90,000
6	Demand function	
7	slope	-2.33
8	intercept	1900
9		
10	**Model**	
11		
12	Revenue	
13	Unit price	$ 500.00
14	Demand	733
15	Number of flights/day	3
16	Total Revenue	$366,666.67
17	Cost	
18	Fixed Cost	$270,000.00
19		
20	Profit	$96,666.67

Solution: Price = $428.57; profit = $115,714.28



Tools for Model Building

- Logic and business principles
- Common mathematical functions
- Data fitting
- Spreadsheet engineering

Logic and Business Principles

- Profit = Revenue - Cost
- Revenue = (Unit price)(Quantity sold)
- Cost = Fixed cost + Unit cost*Quantity produced
- Quantity sold = Min(Quantity produced, Demand)

- Profit = (Unit price)Min(Quantity produced, Demand) – [Fixed cost + (Unit cost)(Quantity produced)]

Modeling Example: Gasoline Consumption

- m = miles/day driven
- d = days/month
- f = miles/gallon

- Miles driven/month = md
- Gallons consumed/month = md/f

Net Present Value

- Measures the worth of a stream of cash flows, taking into account the time value of money.
- A cash flow of F dollars t time periods in the future is worth $F(1 + i)^t$ dollars today, where i is the discount rate.

$$\text{NPV} = \sum_{t=0}^{n} \frac{F_t}{(1+i)^t}$$

- Excel function NPV(*rate*, *value1*, *value2*,...)

Common Mathematical Functions

- *Linear:* $y = mx + b$
- *Logarithmic:* $y = \ln(x)$
- *Polynomial:* $y = ax^2 + bx + c$ *(quadratic)*
- *Power:* $y = ax^b$
- *Exponential:* $y = ab^x$

Data Fitting



Revenue Model

- Total revenue = Price*Sales = Price * (-2.794 * Price + 3149) = -2.794 * Price^2 + 3149 * Price





Spreadsheet Engineering

- Improve the design and format of the spreadsheet itself.
- Improve the process used to develop a spreadsheet.
- Inspect your results carefully and use appropriate tools available in Excel.
  - Use the Data Validation tool
  - Inspect and audit formulas

New Product Development Model

	A	B	C	D	E	F
1	**Moore Pharmaceuticals**					
2						
3	**Data**					
4						
5	Market size	2,000,000				
6	Unit (monthly Rx) revenue	$ 130.00				
7	Unit (monthly Rx) cost	$ 40.00				
8	Discount rate	9%				
9						
10	Project Costs					
11	R&D	$ 700,000,000				
12	Clinical Trials	$ 150,000,000				
13	Total Project Costs	$ 850,000,000				
14						
15	**Model**					
16						
17	Year	1	2	3	4	5
18	Market growth factor		3.00%	3.00%	3.00%	3.00%
19	Market size	2,000,000	2,060,000	2,121,800	2,185,454	2,251,018
20	Market share growth rate		20.00%	20.00%	20.00%	20.00%
21	Market share	8.00%	9.60%	11.52%	13.82%	16.59%
22	Sales	160,000	197,760	244,431	302,117	373,417
23						
24	Annual Revenue	$ 249,600,000	$ 308,505,600	$ 381,312,922	$ 471,302,771	$ 582,530,225
25	Annual Costs	$ 76,800,000	$ 94,924,800	$ 117,327,053	$ 145,016,237	$ 179,240,069
26	Profit	$ 172,800,000	$ 213,580,800	$ 263,985,869	$ 326,286,534	$ 403,290,156
27	Cumulative Net Profit	$(677,200,000)	$(463,619,200)	$(199,633,331)	$ 126,653,203	$ 529,943,358
28						
29	Net Present Value	$ 185,404,860				

New Product Development Model Formulas

	A	B	C	D	E	F
1	**Moore Pharmaceuticals**					
2						
3	**Data**					
4						
5	Market size	2000000				
6	Unit (monthly Rx) revenue	130				
7	Unit (monthly Rx) cost	40				
8	Discount rate	0.09				
9						
10	Project Costs					
11	R&D	700000000				
12	Clinical Trials	150000000				
13	Total Project Costs	=B11+B12				
14						
15	**Model**					
16						
17	Year	1	2	3	4	5
18	Market growth factor		0.03	0.03	0.03	0.03
19	Market size	=B5	=B19*(1+C18)	=C19*(1+D18)	=D19*(1+E18)	=E19*(1+F18)
20	Market share growth rate		0.2	0.2	0.2	0.2
21	Market share	0.08	=B21*(1+C20)	=C21*(1+D20)	=D21*(1+E20)	=E21*(1+F20)
22	Sales	=B19*B21	=C19*C21	=D19*D21	=E19*E21	=F19*F21
23						
24	Annual Revenue	=B22*$B$6*12	=C22*$B$6*12	=D22*$B$6*12	=E22*$B$6*12	=F22*$B$6*12
25	Annual Costs	=B22*$B$7*12	=C22*$B$7*12	=D22*$B$7*12	=E22*$B$7*12	=F22*$B$7*12
26	Profit	=B24-B25	=C24-C25	=D24-D25	=E24-E25	=F24-F25
27	Cumulative Net Profit	=B26-B13	=B27+C26	=C27+D26	=D27+E26	=E27+F26
28						
29	Net Present Value	=NPV(B8,B26:F26)-B13				

Single Period Purchase Decisions (Newsvendor Model)

- C = purchase cost
- R = sale price
- S = salvage value
- D = demand during a single period
- Q = quantity purchased
- Net profit = R * Quantity Sold + S * Surplus Quantity - C * Q

Newsvendor Model Spreadsheet

	A	B
1	**Newsvendor Model**	
2		
3	**Data**	
4		
5	Selling price	18
6	Cost	12
7	Discount price	9
8		
9	**Model**	
10		
11	Demand	41
12	Purchase Quantity	44
13		
14	Quantity Sold	=MIN(B11,B12)
15	Surplus Quantity	=MAX(0,B12-B11)
16		
17	Profit	=B14*B5+B15*B7-B12*B6

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1	**Newsvendor Model**			***Demand***					**Purchase Quantity**					
2				***$ 237.00***	40	41	42	43	44	45	46	47	48	49
3	**Data**			40	$ 240.00	$ 237.00	$ 234.00	$ 231.00	$ 228.00	$ 225.00	$ 222.00	$ 219.00	$ 216.00	$ 213.00
4				41	$ 240.00	$ 246.00	$ 243.00	$ 240.00	$ 237.00	$ 234.00	$ 231.00	$ 228.00	$ 225.00	$ 222.00
5	Selling price	$ 18.00		42	$ 240.00	$ 246.00	$ 252.00	$ 249.00	$ 246.00	$ 243.00	$ 240.00	$ 237.00	$ 234.00	$ 231.00
6	Cost	$ 12.00		43	$ 240.00	$ 246.00	$ 252.00	$ 258.00	$ 255.00	$ 252.00	$ 249.00	$ 246.00	$ 243.00	$ 240.00
7	Discount price	$ 9.00		44	$ 240.00	$ 246.00	$ 252.00	$ 258.00	$ 264.00	$ 261.00	$ 258.00	$ 255.00	$ 252.00	$ 249.00
8				45	$ 240.00	$ 246.00	$ 252.00	$ 258.00	$ 264.00	$ 270.00	$ 267.00	$ 264.00	$ 261.00	$ 258.00
9	**Model**			46	$ 240.00	$ 246.00	$ 252.00	$ 258.00	$ 264.00	$ 270.00	$ 276.00	$ 273.00	$ 270.00	$ 267.00
10				47	$ 240.00	$ 246.00	$ 252.00	$ 258.00	$ 264.00	$ 270.00	$ 276.00	$ 282.00	$ 279.00	$ 276.00
11	Demand	41		48	$ 240.00	$ 246.00	$ 252.00	$ 258.00	$ 264.00	$ 270.00	$ 276.00	$ 282.00	$ 288.00	$ 285.00
12	Purchase Quantity	44		49	$ 240.00	$ 246.00	$ 252.00	$ 258.00	$ 264.00	$ 270.00	$ 276.00	$ 282.00	$ 288.00	$ 294.00
13														
14	Quantity Sold	41												
15	Surplus Quantity	3												
16														
17	Profit	$ 237.00												

Monte Carlo Simulation

- The process of generating random values for uncertain inputs in a model, computing the output variables of interest, and repeating this process for many trials in order to understand the distribution of the output results.

Overbooking Decisions

	A	B
1	**Hotel Overbooking Model**	
2		
3	**Data**	
4		
5	Rooms available	300
6	Price	$120
7	Overbooking cost	$100
8		
9	**Model**	
10		
11	Reservation limit	300
12	Customer demand	290
13	Reservations made	290
14	Cancellations	15
15	Customer arrivals	275
16	Overbooked customers	0
17		
18	Net revenue	$33,000

	A	B
1	**Hotel Overbooking Model**	
2		
3	**Data**	
4		
5	Rooms available	300
6	Price	120
7	Overbooking cost	100
8		
9	**Model**	
10		
11	Reservation limit	300
12	Customer demand	290
13	Reservations made	=MIN(B11,B12)
14	Cancellations	15
15	Customer arrivals	=B13-B14
16	Overbooked customers	=MAX(0,B15-B5)
17		
18	Net revenue	=MIN(B15,B5)*B6-B16*B7

Project Management

TABLE 9.1 Activity and Time Estimate List

	Activity	Predecessors	Activity Time
A	Select steering committee	—	15
B	Develop requirements list	—	50
C	Develop system size estimates	—	20
D	Determine prospective vendors	—	3
E	Form evaluation team	A	7
F	Issue request for proposal	B,C,D,E	6
G	Bidders conference	F	1
H	Review submissions	G	36
I	Select vendor short list	H	6
J	Check vendor references	I	6
K	Vendor demonstrations	I	32
L	User site visit	I	4
M	Select vendor	J,K,L	3
N	Volume-sensitive test	M	15
O	Negotiate contracts	M	18
P	Cost–benefit analysis	N,O	2
Q	Obtain board of directors approval	P	5



Spreadsheet Model

	A	B	C	D	E	F	G	H
1	**Becker Consulting Project Management Model**							
2								
3		***Activity***	***Early***	***Early***	***Latest***	***Latest***		***On Critical***
4	***Activity***	***Time***	***Start***	***Finish***	***Start***	***Finish***	***Slack***	***Path?***
5	A	15.00	0.00	15.00	28.00	43.00	28.00	0
6	B	50.00	0.00	50.00	0.00	50.00	0.00	1
7	C	20.00	0.00	20.00	30.00	50.00	30.00	0
8	D	3.00	0.00	3.00	47.00	50.00	47.00	0
9	E	7.00	15.00	22.00	43.00	50.00	28.00	0
10	F	6.00	50.00	56.00	50.00	56.00	0.00	1
11	G	1.00	56.00	57.00	56.00	57.00	0.00	1
12	H	36.00	57.00	93.00	57.00	93.00	0.00	1
13	I	6.00	93.00	99.00	93.00	99.00	0.00	1
14	J	6.00	99.00	105.00	125.00	131.00	26.00	0
15	K	32.00	99.00	131.00	99.00	131.00	0.00	1
16	L	4.00	99.00	103.00	127.00	131.00	28.00	0
17	M	3.00	131.00	134.00	131.00	134.00	0.00	1
18	N	15.00	134.00	149.00	137.00	152.00	3.00	0
19	O	18.00	134.00	152.00	134.00	152.00	0.00	1
20	P	2.00	152.00	154.00	152.00	154.00	0.00	1
21	Q	5.00	154.00	159.00	154.00	159.00	0.00	1
22								
23		Project completion time		159.00				

Model Formulas

	A	B	C	D	E	F	G	H
1	**Becker Consulting I**							
2								
3		***Activity***	***Early***	***Early***	***Latest***	***Latest***		***On Critical***
4	***Activity***	***Time***	***Start***	***Finish***	***Start***	***Finish***	***Slack***	***Path?***
5	A	15	0	=C5+B5	=F5-B5	=E9	=F5-D5	=IF(G5<0.0001,1,0)
6	B	50	0	=C6+B6	=F6-B6	=E10	=F6-D6	=IF(G6<0.0001,1,0)
7	C	20	0	=C7+B7	=F7-B7	=E10	=F7-D7	=IF(G7<0.0001,1,0)
8	D	3	0	=C8+B8	=F8-B8	=E10	=F8-D8	=IF(G8<0.0001,1,0)
9	E	7	=D5	=C9+B9	=F9-B9	=E10	=F9-D9	=IF(G9<0.0001,1,0)
10	F	6	=MAX(D6,D7,D8,D9)	=C10+B10	=F10-B10	=E11	=F10-D10	=IF(G10<0.0001,1,0)
11	G	1	=D10	=C11+B11	=F11-B11	=E12	=F11-D11	=IF(G11<0.0001,1,0)
12	H	36	=D11	=C12+B12	=F12-B12	=E13	=F12-D12	=IF(G12<0.0001,1,0)
13	I	6	=D12	=C13+B13	=F13-B13	=MIN(E14,E15,E16)	=F13-D13	=IF(G13<0.0001,1,0)
14	J	6	=D13	=C14+B14	=F14-B14	=E17	=F14-D14	=IF(G14<0.0001,1,0)
15	K	32	=D13	=C15+B15	=F15-B15	=E17	=F15-D15	=IF(G15<0.0001,1,0)
16	L	4	=D13	=C16+B16	=F16-B16	=E17	=F16-D16	=IF(G16<0.0001,1,0)
17	M	3	=MAX(D14,D15,D16)	=C17+B17	=F17-B17	=MIN(E18,E19)	=F17-D17	=IF(G17<0.0001,1,0)
18	N	15	=D17	=C18+B18	=F18-B18	=E20	=F18-D18	=IF(G18<0.0001,1,0)
19	O	18	=D17	=C19+B19	=F19-B19	=E20	=F19-D19	=IF(G19<0.0001,1,0)
20	P	2	=MAX(D18,D19)	=C20+B20	=F20-B20	=E21	=F20-D20	=IF(G20<0.0001,1,0)
21	Q	5	=D20	=C21+B21	=F21-B21	=D21	=F21-D21	=IF(G21<0.0001,1,0)
22								
23		Project completion time		=D21				

Model Assumptions, Complexity, and Realism

- All models reflect assumptions used by the modeler.
- Assumptions simplify models and make them easier to manipulate and solve
- Assumptions should be as realistic as necessary to make models useful but not overly complex
- Assumptions should be clearly stated and documented

Example: Retirement Planning

	A	B	C	D	E
1	**Retirement Plan Model**				
2					
3	**Data**				
4	Retirement contribution (% of salary)	8%			
5	Employer match	35%			
6	Annual salary increase	4%			
7	Annual return on investment	8%			
8					
9	**Model**		Employee	Employer	
10	Age	Salary	Contribution	Contribution	Balance
11	25	$50,000	$4,000	$1,400	$5,400
12	26	$ 52,000	$4,160	$1,456	$11,448
13	27	$ 54,080	$4,326	$1,514	$18,204
14	28	$ 56,243	$4,499	$1,575	$25,735
15	29	$ 58,493	$4,679	$1,638	$34,111
16	30	$ 60,833	$4,867	$1,703	$43,410
17	31	$ 63,266	$5,061	$1,771	$53,715
18	32	$ 65,797	$5,264	$1,842	$65,119
19	33	$ 68,428	$5,474	$1,916	$77,719
20	34	$ 71,166	$5,693	$1,993	$91,622
21	35	$ 74,012	$5,921	$2,072	$106,945
22	36	$ 76,973	$6,158	$2,155	$123,814
23	37	$ 80,052	$6,404	$2,241	$142,364
24	38	$ 83,254	$6,660	$2,331	$162,745
25	39	$ 86,584	$6,927	$2,424	$185,115
26	40	$ 90,047	$7,204	$2,521	$209,650
27	41	$ 93,649	$7,492	$2,622	$236,536
28	42	$ 97,395	$7,792	$2,727	$265,977
29	43	$ 101,291	$8,103	$2,836	$298,195
30	44	$ 105,342	$8,427	$2,950	$333,428
31	45	$ 109,556	$8,764	$3,068	$371,934
32	46	$ 113,938	$9,115	$3,190	$413,994
33	47	$ 118,496	$9,480	$3,318	$459,911
34	48	$ 123,236	$9,859	$3,451	$510,013
35	49	$ 128,165	$10,253	$3,589	$564,656
36	50	$ 133,292	$10,663	$3,732	$624,224

	A	B	C	D	E
1	**Retirement Plan Model**				
2					
3	**Data**				
4	Retirement contribution (% of salary)	0.08			
5	Employer match	0.35			
6	Annual salary increase	0.04			
7	Annual return on investment	0.08			
8					
9	**Model**		Employee	Employer	
10	Age	Salary	Contribution	Contribution	Balance
11	25	50000	=B11*$B$4	=$B$5*C11	=C11+D11
12	26	= B11*(1+$B$6)	=B12*$B$4	=$B$5*C12	=E11*(1+$B$7) + C12+D12
13	27	= B12*(1+$B$6)	=B13*$B$4	=$B$5*C13	=E12*(1+$B$7) + C13+D13
14	28	= B13*(1+$B$6)	=B14*$B$4	=$B$5*C14	=E13*(1+$B$7) + C14+D14